
BB.Servo.PCA9685
BB integration for driving RC servos via PCA9685 16-channel PWM controller over I2C.
This library provides a controller and actuator module for controlling RC servos
connected to a PCA9685 board.
Installation
Add bb_servo_pca9685 to your list of dependencies in mix.exs:
def deps do
  [
    {:bb_servo_pca9685, "~> 0.5.0"}
  ]
end
Requirements
	PCA9685 PWM controller connected via I2C
	BB framework (~> 0.2)

Usage
Define a controller and joints with servo actuators in your robot DSL:
defmodule MyRobot do
  use BB

  # Define the PCA9685 controller at robot level
  controller :pca9685, {BB.Servo.PCA9685.Controller, bus: "i2c-1", address: 0x40}

  topology do
    link :base do
      joint :shoulder, type: :revolute do
        limit lower: ~u(-45 degree), upper: ~u(45 degree), velocity: ~u(60 degree_per_second)

        actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 0, controller: :pca9685}
        sensor :feedback, {BB.Sensor.OpenLoopPositionEstimator, actuator: :servo}

        link :upper_arm do
          joint :elbow, type: :revolute do
            limit lower: ~u(-90 degree), upper: ~u(90 degree), velocity: ~u(60 degree_per_second)

            actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 1, controller: :pca9685}
            sensor :feedback, {BB.Sensor.OpenLoopPositionEstimator, actuator: :servo}

            link :forearm do
            end
          end
        end
      end
    end
  end
end
The actuator automatically derives its configuration from the joint limits - no
need to specify servo rotation range or speed separately.
Sending Commands
Use the BB.Actuator module to send commands to servos. Three delivery methods
are available:
Pubsub Delivery (for orchestration)
Commands are published via pubsub, enabling logging, replay, and multi-subscriber
patterns:
# Send position command via pubsub
BB.Actuator.set_position(MyRobot, [:base, :shoulder, :servo], 0.5)

# With options
BB.Actuator.set_position(MyRobot, [:base, :shoulder, :servo], 0.5,
  command_id: make_ref()
)
Direct Delivery (for time-critical control)
Commands bypass pubsub for lower latency. Use when responsiveness matters more
than observability:
# Fire-and-forget
BB.Actuator.set_position!(MyRobot, :servo, 0.5)
Synchronous Delivery (with acknowledgement)
Wait for the actuator to acknowledge the command:
case BB.Actuator.set_position_sync(MyRobot, :servo, 0.5) do
  {:ok, :accepted} -> :ok
  {:error, reason} -> handle_error(reason)
end
Components
Controller
BB.Servo.PCA9685.Controller manages communication with the PCA9685 board.
Define one controller per physical PCA9685 device.
Options:
	Option	Type	Default	Description
	bus	string	required	I2C bus name (e.g. "i2c-1")
	address	integer	0x40	I2C address of the PCA9685
	frequency	integer	50	PWM frequency in Hz
	oe_pin	integer	nil	Optional output-enable GPIO pin

Actuator
BB.Servo.PCA9685.Actuator controls a single servo on one of the 16 channels.
Options:
	Option	Type	Default	Description
	channel	0-15	required	PCA9685 channel number
	controller	atom	required	Name of the controller in robot registry
	min_pulse	integer	500	Minimum PWM pulse width (microseconds)
	max_pulse	integer	2500	Maximum PWM pulse width (microseconds)
	reverse?	boolean	false	Reverse rotation direction

Behaviour:
	Maps joint position limits directly to PWM range
	Clamps commanded positions to joint limits
	Publishes BB.Message.Actuator.BeginMotion after each command
	Calculates expected arrival time based on joint velocity limit

Sensor
Use BB.Sensor.OpenLoopPositionEstimator from the BB core library for position
feedback. It subscribes to actuator BeginMotion messages and interpolates
position during movement.
sensor :feedback, {BB.Sensor.OpenLoopPositionEstimator, actuator: :servo}
How It Works
Architecture
Controller (GenServer)
    |
    v wraps
PCA9685.Device (I2C communication)
    ^
    | used by
Actuator (GenServer) --publishes--> BeginMotion --> Sensor (GenServer)
                                                        |
                                                        v publishes
                                                    JointState
Multiple actuators share a single controller. Each actuator controls one of the
16 available channels.
Position Mapping
The actuator maps the joint's position limits to the servo's PWM range:
Joint lower limit  ->  min_pulse (500 microseconds)
Joint upper limit  ->  max_pulse (2500 microseconds)
Joint centre       ->  mid_pulse (1500 microseconds)
For a joint with limits -45 degrees to +45 degrees:
	-45 degrees maps to 500 microseconds
	0 degrees maps to 1500 microseconds
	+45 degrees maps to 2500 microseconds

Position Feedback
Since RC servos don't provide position feedback, the open-loop position
estimator estimates position based on commanded targets and expected arrival
times:
	Actuator sends command and publishes BeginMotion with expected arrival time
	Sensor receives BeginMotion and interpolates position during movement
	After arrival time, sensor reports the target position

This provides realistic position feedback for trajectory planning and monitoring.
Motion Lifecycle
When a position command is processed:
	Actuator clamps position to joint limits
	Converts angle to PWM pulse width
	Sends command to controller via BB.Process.call
	Controller writes PWM to the PCA9685 over I2C
	Publishes BB.Message.Actuator.BeginMotion with:	initial_position - where the servo was
	target_position - where it's going
	expected_arrival - when it should arrive (monotonic milliseconds)
	command_id - correlation ID (if provided)
	command_type - :position



Documentation
Full documentation is available at HexDocs.


  

    Change Log

All notable changes to this project will be documented in this file.
See Conventional Commits for commit guidelines.
v0.5.0 (2026-01-11)
Features:
	migrate to structured error system (#8) by James Harton

Improvements:
	use structured error for missing OE pin (#9) by James Harton

v0.4.0 (2025-12-24)
Breaking Changes:
	update to bb 0.8 wrapper GenServer pattern (#7) by James Harton

v0.3.1 (2025-12-20)
Improvements
	update for compatibility with BB 0.6.

v0.3.0 (2025-12-14)
Features:
	implement BB.Safety behaviour for safe disarm (#3) by James Harton

v0.2.0 (2025-12-13)
Features:
	add PCA9685 servo controller, actuator, and sensor modules by James Harton

Improvements:
	use standard actuator command interface by James Harton



  

    Getting Started

This guide walks you through setting up a PCA9685 PWM controller to drive RC
servos with BB.Servo.PCA9685.
Hardware Requirements
	Raspberry Pi (or any board with I2C support)
	PCA9685 16-channel PWM controller board
	RC servos (standard hobby servos with 3-wire connectors)
	5V power supply for the servos
	Jumper wires

Understanding the PCA9685
The PCA9685 is a 16-channel, 12-bit PWM controller that communicates via I2C.
Key features:
	16 independent PWM channels - Control up to 16 servos per board
	I2C interface - Only uses 2 GPIO pins regardless of servo count
	Chainable - Connect multiple boards for more channels (up to 62 boards)
	Hardware PWM - Precise timing without CPU load

Board Pinout
PCA9685 Board
┌─────────────────────────────────────────┐
│  V+  VCC  SDA  SCL  GND  OE            │  ← Control side
├─────────────────────────────────────────┤
│  PWM0  PWM1  PWM2  ... PWM15           │  ← Servo outputs
│  (each has 3 pins: V+, GND, PWM)       │
└─────────────────────────────────────────┘
Wiring
I2C Connection
	PCA9685	Raspberry Pi
	VCC	3.3V (Pin 1)
	GND	GND (Pin 6)
	SDA	SDA (Pin 3, GPIO 2)
	SCL	SCL (Pin 5, GPIO 3)

Servo Power
Important: Servos draw significant current. Do NOT power servos from the
Pi's 5V pin. Use an external 5V supply.

	Connection	Description
	V+ terminal	External 5V power supply (+)
	GND	External 5V power supply (-) AND Pi GND

Servo Connection
Connect each servo to its channel (0-15):
	Servo Wire	PCA9685 Channel Pin
	Brown/Black (GND)	GND row
	Red (Power)	V+ row
	Orange/Yellow (Signal)	PWM row

Wiring Diagram
                    External 5V Supply
                         │
                    ┌────┴────┐
                    │  + -    │
                    └────┬────┘
                         │
Raspberry Pi         PCA9685              Servos
───────────         ────────              ──────
3.3V ──────────────► VCC
GND ───────────────► GND ◄──── GND ◄───── All servo GND
SDA (GPIO 2) ──────► SDA                  (brown wire)
SCL (GPIO 3) ──────► SCL
                     V+ ◄───── +5V ◄───── All servo power
                                          (red wire)
                     PWM0 ────────────────► Servo 0 signal
                     PWM1 ────────────────► Servo 1 signal
                     ...
                     PWM15 ───────────────► Servo 15 signal
Software Setup
1. Enable I2C on Raspberry Pi
# Using raspi-config
sudo raspi-config
# Navigate to: Interface Options → I2C → Enable

# Or edit config directly
echo "dtparam=i2c_arm=on" | sudo tee -a /boot/config.txt
sudo reboot

Verify I2C is enabled:
ls /dev/i2c*
# Should show: /dev/i2c-1

2. Install I2C Tools (Optional but Recommended)
sudo apt-get update
sudo apt-get install i2c-tools

Scan for the PCA9685:
i2cdetect -y 1

You should see the device at address 0x40 (default):
     0  1  2  3  4  5  6  7  8  9  a  b  c  d  e  f
00:                         -- -- -- -- -- -- -- --
10: -- -- -- -- -- -- -- -- -- -- -- -- -- -- -- --
20: -- -- -- -- -- -- -- -- -- -- -- -- -- -- -- --
30: -- -- -- -- -- -- -- -- -- -- -- -- -- -- -- --
40: 40 -- -- -- -- -- -- -- -- -- -- -- -- -- -- --
50: -- -- -- -- -- -- -- -- -- -- -- -- -- -- -- --
60: -- -- -- -- -- -- -- -- -- -- -- -- -- -- -- --
70: 70 -- -- -- -- -- -- --
3. Add Dependencies
Add bb_servo_pca9685 to your mix.exs:
def deps do
  [
    {:bb, "~> 0.2"},
    {:bb_servo_pca9685, "~> 0.1"}
  ]
end
Then fetch dependencies:
mix deps.get

Verify Your Setup
Create a simple test to verify everything works:
# In IEx on your Raspberry Pi
iex> {:ok, device} = PCA9685.acquire(bus: "i2c-1", address: 0x40)
{:ok, #PID<0.123.0>}

iex> PCA9685.Device.pulse_width(device, 0, 1500)
:ok
This should move the servo on channel 0 to its centre position. If you see
:ok, your setup is working correctly.
Release the device when done:
iex> PCA9685.release(device)
:ok
Multiple PCA9685 Boards
To use more than 16 servos, connect multiple PCA9685 boards to the same I2C bus
with different addresses. Change the address using the solder jumpers on the
board:
	Jumpers	Address
	None	0x40
	A0	0x41
	A1	0x42
	A0 + A1	0x43
	...	...
	All (A0-A5)	0x7F

Troubleshooting
"No such file or directory" for /dev/i2c-1
I2C is not enabled:
sudo raspi-config
# Interface Options → I2C → Enable
sudo reboot

Device not found at expected address
	Check wiring connections
	Verify VCC is connected to 3.3V (not 5V)
	Run i2cdetect -y 1 to scan for devices
	Check address jumpers on the board

Servo doesn't move
	Check servo power supply is connected
	Verify the channel number (0-15)
	Try a different channel
	Check servo wiring (signal to PWM row)

Servo jitters or moves erratically
	Use a separate power supply for servos (not the Pi's 5V)
	Add capacitors across the servo power rails
	Ensure ground is shared between Pi, PCA9685, and power supply

I2C communication errors
	Keep I2C wires short (under 50cm)
	Check for loose connections
	Try reducing I2C speed if using long wires

Next Steps
Now that your hardware is set up, proceed to Basic Usage to
learn how to integrate servos into your BB robot.


  

    Basic Usage

This tutorial shows you how to define a PCA9685 controller and servo-controlled
joints in your BB robot.
Prerequisites
	Completed Getting Started
	PCA9685 connected via I2C
	At least one servo connected to channel 0

Defining a Robot with PCA9685 Servos
Create a robot module with a controller and servo-controlled joints:
defmodule MyRobot do
  use BB.Robot

  robot do
    # Define the PCA9685 controller
    controller :pca9685, {BB.Servo.PCA9685.Controller,
      bus: "i2c-1",
      address: 0x40
    }

    link :base do
      joint :pan, type: :revolute do
        # Define the joint's motion limits
        limit lower: ~u(-90 degree),
              upper: ~u(90 degree),
              velocity: ~u(60 degree_per_second)

        # Attach the servo actuator on channel 0
        actuator :servo, {BB.Servo.PCA9685.Actuator,
          channel: 0,
          controller: :pca9685
        }

        link :head do
          # Child links go here
        end
      end
    end
  end
end
Understanding the Configuration
Controller Options
The controller manages the I2C connection to the PCA9685:
controller :pca9685, {BB.Servo.PCA9685.Controller,
  bus: "i2c-1",         # Required: I2C bus name
  address: 0x40,        # Required: I2C address (default for PCA9685)
  pwm_freq: 50,         # Optional: PWM frequency in Hz (default: 50)
  oe_pin: 25            # Optional: GPIO pin for output enable
}
	bus - The I2C bus device name (usually "i2c-1" on Raspberry Pi)
	address - The I2C address of the PCA9685 (default 0x40)
	pwm_freq - PWM frequency, 50 Hz is standard for servos
	oe_pin - Optional GPIO pin connected to the PCA9685's OE (Output Enable) pin

Joint Limits
The limit block defines the physical constraints of your joint:
	lower - Minimum position (maps to servo's minimum pulse)
	upper - Maximum position (maps to servo's maximum pulse)
	velocity - Maximum rotation speed (used for timing calculations)

These values are used by the actuator to:
	Map positions to PWM pulse widths
	Clamp commanded positions to safe values
	Calculate expected movement duration

Actuator Options
The actuator controls a single servo channel:
actuator :servo, {BB.Servo.PCA9685.Actuator,
  channel: 0,          # Required: PCA9685 channel (0-15)
  controller: :pca9685, # Required: name of the controller
  min_pulse: 500,      # Optional: minimum pulse width in µs (default: 500)
  max_pulse: 2500,     # Optional: maximum pulse width in µs (default: 2500)
  reverse?: false      # Optional: reverse rotation direction (default: false)
}
Most servos work well with the defaults. Adjust min_pulse and max_pulse if
your servo has different endpoints.
Starting the Robot
Start your robot in your application supervision tree:
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [
      MyRobot
    ]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end
end
Or start it manually in IEx:
iex> MyRobot.start_link()
{:ok, #PID<0.123.0>}
Commanding the Servo
Send position commands to the actuator:
# Move to centre (0 degrees)
BB.Actuator.set_position(MyRobot, :servo, 0.0)

# Move to -45 degrees (in radians)
BB.Actuator.set_position(MyRobot, :servo, -0.785)

# Using the unit sigil for degrees
import BB.Unit
BB.Actuator.set_position(MyRobot, :servo, ~u(-45 degree) |> BB.Robot.Units.to_radians())
Note: BB uses radians internally. Convert degrees to radians when sending
commands, or use the unit conversion functions.

Position Clamping
The actuator automatically clamps positions to the joint limits:
# Joint limits are -90° to +90°
# This command will be clamped to +90° (π/2 radians)
BB.Actuator.set_position(MyRobot, :servo, 3.14)  # Requested: 180°, actual: 90°
Reversing Direction
If your servo rotates in the opposite direction to what you expect, use the
reverse? option:
actuator :servo, {BB.Servo.PCA9685.Actuator,
  channel: 0,
  controller: :pca9685,
  reverse?: true
}
This inverts the PWM mapping so that:
	Lower limit → maximum pulse
	Upper limit → minimum pulse

Example: Pan-Tilt Head
Here's a complete example with two servos for a pan-tilt mechanism:
defmodule PanTiltRobot do
  use BB.Robot

  robot do
    controller :pca9685, {BB.Servo.PCA9685.Controller,
      bus: "i2c-1",
      address: 0x40
    }

    link :base do
      joint :pan, type: :revolute do
        limit lower: ~u(-90 degree), upper: ~u(90 degree), velocity: ~u(90 degree_per_second)
        actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 0, controller: :pca9685}

        link :pan_platform do
          joint :tilt, type: :revolute do
            limit lower: ~u(-45 degree), upper: ~u(45 degree), velocity: ~u(60 degree_per_second)
            actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 1, controller: :pca9685}

            link :camera_mount do
              # Camera attached here
            end
          end
        end
      end
    end
  end
end
Command both servos:
# Look left and up
BB.Actuator.set_position(PanTiltRobot, :pan, -0.785)   # -45°
BB.Actuator.set_position(PanTiltRobot, :tilt, 0.524)   # +30°
Example: Hexapod Leg (6 Servos)
The PCA9685's 16 channels make it ideal for multi-servo robots:
defmodule HexapodLeg do
  use BB.Robot

  robot do
    controller :pca9685, {BB.Servo.PCA9685.Controller,
      bus: "i2c-1",
      address: 0x40
    }

    link :body do
      # Leg 1
      joint :leg1_coxa, type: :revolute do
        limit lower: ~u(-45 degree), upper: ~u(45 degree), velocity: ~u(90 degree_per_second)
        actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 0, controller: :pca9685}

        link :leg1_coxa_link do
          joint :leg1_femur, type: :revolute do
            limit lower: ~u(-90 degree), upper: ~u(30 degree), velocity: ~u(90 degree_per_second)
            actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 1, controller: :pca9685}

            link :leg1_femur_link do
              joint :leg1_tibia, type: :revolute do
                limit lower: ~u(-120 degree), upper: ~u(0 degree), velocity: ~u(90 degree_per_second)
                actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 2, controller: :pca9685}

                link :leg1_foot do
                end
              end
            end
          end
        end
      end

      # Leg 2 uses channels 3, 4, 5
      # Leg 3 uses channels 6, 7, 8
      # ... and so on
    end
  end
end
Multiple PCA9685 Boards
For robots with more than 16 servos, define multiple controllers:
defmodule BigRobot do
  use BB.Robot

  robot do
    # First board at default address
    controller :pca9685_a, {BB.Servo.PCA9685.Controller,
      bus: "i2c-1",
      address: 0x40
    }

    # Second board with A0 jumper set
    controller :pca9685_b, {BB.Servo.PCA9685.Controller,
      bus: "i2c-1",
      address: 0x41
    }

    link :base do
      # First 16 servos use :pca9685_a
      joint :joint_0, type: :revolute do
        limit lower: ~u(-90 degree), upper: ~u(90 degree), velocity: ~u(60 degree_per_second)
        actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 0, controller: :pca9685_a}
        link :link_0 do end
      end

      # Servos 17+ use :pca9685_b
      joint :joint_16, type: :revolute do
        limit lower: ~u(-90 degree), upper: ~u(90 degree), velocity: ~u(60 degree_per_second)
        actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 0, controller: :pca9685_b}
        link :link_16 do end
      end
    end
  end
end
Output Enable Control
If you've connected the PCA9685's OE pin to a GPIO, you can enable/disable all
outputs:
controller :pca9685, {BB.Servo.PCA9685.Controller,
  bus: "i2c-1",
  address: 0x40,
  oe_pin: 25  # GPIO 25 connected to OE
}
Control outputs via the controller:
# Disable all servo outputs (servos go limp)
BB.Process.call(MyRobot, :pca9685, :output_disable)

# Re-enable outputs
BB.Process.call(MyRobot, :pca9685, :output_enable)
This is useful for:
	Emergency stops
	Allowing manual positioning of servos
	Reducing power consumption when idle

Next Steps
To get position feedback from your servos, see Position Feedback.


  

    Position Feedback

RC servos don't provide position feedback, but BB.Servo.PCA9685.Sensor can
estimate position based on commanded targets and timing. This tutorial shows you
how to set up and use position feedback.
Prerequisites
	Completed Basic Usage
	A working servo joint with PCA9685 controller

Adding a Feedback Sensor
Add the sensor to your joint definition:
defmodule MyRobot do
  use BB.Robot

  robot do
    controller :pca9685, {BB.Servo.PCA9685.Controller,
      bus: "i2c-1",
      address: 0x40
    }

    link :base do
      joint :pan, type: :revolute do
        limit lower: ~u(-90 degree), upper: ~u(90 degree), velocity: ~u(60 degree_per_second)

        actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 0, controller: :pca9685}
        sensor :feedback, {BB.Servo.PCA9685.Sensor, actuator: :servo}

        link :head do
        end
      end
    end
  end
end
The sensor requires the actuator option to know which actuator to subscribe to.
How Position Feedback Works
Since RC servos don't report their actual position, the sensor estimates it:
	Actuator publishes commands - When you call set_position, the actuator
publishes a PositionCommand message with the target angle and expected
arrival time

	Sensor subscribes - The sensor receives these messages and tracks the
target position and expected arrival time

	Position interpolation - During movement, the sensor interpolates between
the previous position and target based on elapsed time

	JointState publishing - The sensor publishes JointState messages with
the estimated position


Timeline Example
Time 0ms:    Command sent (target: 45°, arrival: 500ms)
             Sensor receives command, starts interpolating

Time 100ms:  Estimated position: 9° (20% of the way)
Time 250ms:  Estimated position: 22.5° (50% of the way)
Time 500ms:  Estimated position: 45° (arrived)

Time 600ms:  Position stable at 45° (no change published)
Time 5000ms: Sync publish at 45° (max_silence reached)
Sensor Options
sensor :feedback, {BB.Servo.PCA9685.Sensor,
  actuator: :servo,           # Required: actuator to subscribe to
  publish_rate: ~u(50 hertz), # Optional: how often to check for changes (default: 50 Hz)
  max_silence: ~u(5 second)   # Optional: max time between publishes (default: 5s)
}
publish_rate
How often the sensor checks for position changes. Higher rates give smoother
feedback during movement but use more resources.
	~u(50 hertz) - Default, good for most applications
	~u(100 hertz) - Smoother feedback for fast movements
	~u(10 hertz) - Lower resource usage for slow-moving joints

max_silence
Even when the position hasn't changed, the sensor publishes periodically to keep
subscribers in sync. This handles:
	Late subscribers that missed earlier updates
	Recovery from dropped messages
	Monitoring systems that expect regular updates

Set to a higher value if you want less traffic when idle.
Subscribing to Position Updates
Subscribe to the sensor's JointState messages:
# Subscribe to the sensor topic
BB.subscribe(MyRobot, [:sensor, :pan, :feedback])

# In your GenServer or process
def handle_info(%BB.Message{payload: %BB.Message.Sensor.JointState{} = joint_state}, state) do
  [position] = joint_state.positions
  IO.puts("Pan position: #{position} radians")
  {:noreply, state}
end
Reading Current Position
You can also query the robot's state directly:
# Get current joint positions
state = BB.Robot.State.get(MyRobot)
pan_position = BB.Robot.State.get_joint_position(state, :pan)
Example: Position Logger
Here's a complete example that logs position changes:
defmodule PositionLogger do
  use GenServer

  def start_link(robot) do
    GenServer.start_link(__MODULE__, robot, name: __MODULE__)
  end

  def init(robot) do
    BB.subscribe(robot, [:sensor, :pan, :feedback])
    {:ok, %{robot: robot}}
  end

  def handle_info(%BB.Message{payload: %BB.Message.Sensor.JointState{} = js}, state) do
    [position] = js.positions
    degrees = position * 180 / :math.pi()
    IO.puts("[#{DateTime.utc_now()}] Pan: #{Float.round(degrees, 1)}°")
    {:noreply, state}
  end
end
Start the logger:
{:ok, _} = MyRobot.start_link()
{:ok, _} = PositionLogger.start_link(MyRobot)

# Move the servo and watch the logs
BB.Actuator.set_position(MyRobot, :servo, 0.785)
# Output:
# [2025-01-15 10:30:00.000000Z] Pan: 9.0°
# [2025-01-15 10:30:00.020000Z] Pan: 18.0°
# [2025-01-15 10:30:00.040000Z] Pan: 27.0°
# ... (interpolated positions during movement)
# [2025-01-15 10:30:00.500000Z] Pan: 45.0°
Example: Wait for Movement Complete
Wait for the servo to reach its target position:
defmodule ServoHelper do
  def move_and_wait(robot, actuator, target, timeout \\ 5000) do
    # Subscribe to sensor updates
    BB.subscribe(robot, [:sensor, :pan, :feedback])

    # Send the command
    BB.Actuator.set_position(robot, actuator, target)

    # Wait for position to match target
    wait_for_position(target, timeout)
  end

  defp wait_for_position(target, timeout) do
    receive do
      %BB.Message{payload: %BB.Message.Sensor.JointState{positions: [position]}}
      when abs(position - target) < 0.01 ->
        :ok
    after
      timeout -> {:error, :timeout}
    end
  end
end

# Usage
:ok = ServoHelper.move_and_wait(MyRobot, :servo, 0.785)
IO.puts("Servo reached target!")
Example: Multi-Joint Position Monitor
Monitor all servos in a hexapod leg:
defmodule LegMonitor do
  use GenServer

  @joints [:leg1_coxa, :leg1_femur, :leg1_tibia]

  def start_link(robot) do
    GenServer.start_link(__MODULE__, robot, name: __MODULE__)
  end

  def init(robot) do
    # Subscribe to all leg joint sensors
    for joint <- @joints do
      BB.subscribe(robot, [:sensor, joint, :feedback])
    end

    {:ok, %{robot: robot, positions: %{}}}
  end

  def handle_info(%BB.Message{frame_id: joint, payload: %BB.Message.Sensor.JointState{} = js}, state) do
    [position] = js.positions
    positions = Map.put(state.positions, joint, position)

    # Print all positions when we have updates for all joints
    if map_size(positions) == length(@joints) do
      print_leg_position(positions)
    end

    {:noreply, %{state | positions: positions}}
  end

  defp print_leg_position(positions) do
    coxa = rad_to_deg(positions[:leg1_coxa])
    femur = rad_to_deg(positions[:leg1_femur])
    tibia = rad_to_deg(positions[:leg1_tibia])

    IO.puts("Leg position: coxa=#{coxa}° femur=#{femur}° tibia=#{tibia}°")
  end

  defp rad_to_deg(rad), do: Float.round(rad * 180 / :math.pi(), 1)
end
Limitations
Remember that this is estimated position, not actual position:
	The servo might not reach the target (blocked, insufficient torque)
	The servo might overshoot or oscillate
	The timing might not match the real servo exactly

For applications requiring precise position feedback, consider:
	Adding a physical sensor (potentiometer, encoder) to your servo
	Using a servo with built-in feedback (smart servos)
	Using stepper motors with encoders

Comparison with BB.Servo.Pigpio
The sensor works identically to BB.Servo.Pigpio.Sensor. If you're migrating
from pigpio to PCA9685, simply:
	Replace BB.Servo.Pigpio.Sensor with BB.Servo.PCA9685.Sensor
	Ensure the actuator option points to a PCA9685 actuator

The message types and behaviour are the same.
Next Steps
You now have a complete servo setup with position feedback. Explore the BB
framework documentation to learn about:
	Trajectory planning for smooth multi-joint movements
	Inverse kinematics for end-effector positioning
	State machines for complex behaviours



  

    
BB.Servo.PCA9685 
    



      
BB integration for driving RC servos via the PCA9685 PWM controller.
This library provides controller, actuator and sensor modules for controlling
RC servos through a PCA9685 16-channel PWM controller connected via I2C.
Components
	BB.Servo.PCA9685.Controller - Manages the PCA9685 device connection
	BB.Servo.PCA9685.Actuator - Controls servo position via a controller channel
	BB.Servo.PCA9685.Sensor - Provides position feedback by subscribing to actuator commands

Requirements
	PCA9685 PWM controller connected via I2C
	The pca9685 library for communication with the device

Quick Start
Define a controller and joints with servo actuators in your robot DSL:
controller :pca9685, {BB.Servo.PCA9685.Controller, bus: "i2c-1", address: 0x40}

joint :shoulder, type: :revolute do
  limit lower: ~u(-45 degree), upper: ~u(45 degree), velocity: ~u(60 degree_per_second)

  actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 0, controller: :pca9685}
  sensor :feedback, {BB.Servo.PCA9685.Sensor, actuator: :servo}
end

joint :elbow, type: :revolute do
  limit lower: ~u(-90 degree), upper: ~u(90 degree), velocity: ~u(45 degree_per_second)

  actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 1, controller: :pca9685}
  sensor :feedback, {BB.Servo.PCA9685.Sensor, actuator: :servo}
end
The actuator automatically derives its configuration from the joint limits - no need
to specify servo rotation range or speed separately.
How It Works
Controller
The controller wraps a PCA9685.Device process and provides a stable reference
for actuators. Multiple actuators can share a single controller, each using a
different channel (0-15). The controller:
	Manages the I2C connection to the PCA9685
	Sets the PWM frequency (default 50Hz for servos)
	Optionally controls an output enable pin

Actuator
The actuator maps the joint's position limits directly to the servo's PWM range:
	Joint lower limit → minimum pulse width (default 500µs)
	Joint upper limit → maximum pulse width (default 2500µs)
	Centre position calculated as midpoint of limits

When commanded to a position, the actuator:
	Clamps the position to joint limits
	Converts to PWM pulse width
	Sends command to the controller
	Publishes PositionCommand for sensors

Sensor
The sensor subscribes to actuator position commands and publishes JointState
messages. It provides:
	Position interpolation during movement
	Configurable publish rate (default 50Hz)
	Periodic sync publishing even when idle (default every 5 seconds)

Multiple PCA9685 Boards
For robots with more than 16 servos, you can define multiple controllers:
controller :pca9685_a, {BB.Servo.PCA9685.Controller, bus: "i2c-1", address: 0x40}
controller :pca9685_b, {BB.Servo.PCA9685.Controller, bus: "i2c-1", address: 0x41}

joint :shoulder, type: :revolute do
  actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 0, controller: :pca9685_a}
  # ...
end

joint :gripper, type: :revolute do
  actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 0, controller: :pca9685_b}
  # ...
end

      




  

    
BB.Servo.PCA9685.Actuator 
    



      
An actuator that uses a PCA9685 controller to drive a servo.
This actuator derives its configuration from the joint constraints defined in the robot:
	Position limits from joint.limits.lower and joint.limits.upper
	Velocity limit from joint.limits.velocity
	PWM range maps linearly to the joint's position range

When a position command is received, the actuator:
	Clamps the position to joint limits
	Converts to PWM pulse width
	Sends PWM command to the PCA9685 controller
	Publishes a BB.Message.Actuator.BeginMotion for sensors to consume

Example DSL Usage
controller :pca9685, {BB.Servo.PCA9685.Controller, bus: "i2c-1", address: 0x40}

joint :shoulder, type: :revolute do
  limit lower: ~u(-45 degree), upper: ~u(45 degree), velocity: ~u(60 degree_per_second)

  actuator :servo, {BB.Servo.PCA9685.Actuator, channel: 0, controller: :pca9685}
  sensor :feedback, {BB.Sensor.OpenLoopPositionEstimator, actuator: :servo}
end

        disarm(opts)

      


        Disable the servo by setting pulse width to 0.

Disable the servo by setting pulse width to 0.
Called by BB.Safety.Controller when the robot is disarmed or crashes.
This function works without GenServer state - it receives the robot module,
controller name, and channel from the opts provided during registration.

  


        

      


  

    
BB.Servo.PCA9685.Controller 
    



      
A controller that manages a PCA9685 PWM device.
This controller wraps a PCA9685.Device process and provides an interface
for actuators to set servo pulse widths. Multiple actuators can share a
single controller, with each actuator controlling a different channel (0-15).
Configuration
The controller is typically defined in the robot DSL:
controller :pca9685, {BB.Servo.PCA9685.Controller,
  bus: "i2c-1",
  address: 0x40,
  pwm_freq: 50
}
Options
	:bus - (required) The I2C bus name, e.g., "i2c-1"
	:address - (required) The I2C address of the PCA9685, e.g., 0x40
	:pwm_freq - PWM frequency in Hz (default: 50, suitable for servos)
	:oe_pin - Optional GPIO pin for output enable control

Safety
This controller implements the BB.Controller behaviour. When the robot is disarmed
or crashes, the OE pin is pulled high to disable all servo outputs. If no OE pin
is configured, the disarm callback returns :ok (individual actuators handle
their own channel disarm).

      


      
        Summary


  
    Functions
  


    
      
        disarm(opts)

      


        Disable all servo outputs by pulling the OE pin high.

Disable all servo outputs by pulling the OE pin high.
Called by BB.Safety.Controller when the robot is disarmed or crashes.
If no OE pin is configured, returns :ok (individual actuators handle their channels).
Note: This callback calls through the Controller GenServer, so it only works
if the Controller process is still alive. For crash scenarios, individual
Actuator disarm callbacks provide per-channel safety.

  


        

      


  

    
TestRobot 
    



      
Stub robot module for testing.

BB.Error.Hardware.NoOutputEnablePin exception
    



      
Output enable pin not configured.
The PCA9685 controller was asked to enable or disable outputs, but no
output enable (OE) GPIO pin was configured. Without an OE pin, the controller
cannot globally enable or disable all servo outputs.
Individual actuators can still be disabled by setting their pulse width to 0.

    Functions
  


    
      
        exception(args)

      


        Create an Elixir.BB.Error.Hardware.NoOutputEnablePin without raising it.

          @type t() :: %BB.Error.Hardware.NoOutputEnablePin{
  __exception__: true,
  bread_crumbs: term(),
  class: term(),
  controller: atom() | nil,
  path: term(),
  splode: term(),
  stacktrace: term(),
  vars: term()
}

          @spec exception(opts :: Keyword.t()) :: %BB.Error.Hardware.NoOutputEnablePin{
  __exception__: true,
  bread_crumbs: term(),
  class: term(),
  controller: term(),
  path: term(),
  splode: term(),
  stacktrace: term(),
  vars: term()
}


      


Create an Elixir.BB.Error.Hardware.NoOutputEnablePin without raising it.
Keys
	:controller

